
Subject: Re: azimuthal median

Posted by [Wox](#) on Tue, 15 Jan 2008 09:48:12 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Mon, 14 Jan 2008 14:22:10 +0100, Bringfried Stecklum
<stecklum@tls-tautenburg.de> wrote:

> Hi folks,

>

> I am looking for a faster way to compute the azimuthal median in
> dependence on radius than the brute-force method, i.e. getting the index
> of pixels within a certain annulus, and using median(image[idx]). Is
> this another case for the histogram wizards out there?

>

> regards,

>

> Bringfried

I'm not familiar azimuthal median but what about image warping:

```
; Make azimuthal range
```

```
a0=0.
```

```
a1=2*!pi
```

```
ai=0.1
```

```
na=ceil((a1-a0)/ai)+1
```

```
ai=(a1-a0)/(na-1)
```

```
a=a0+ai*indgen(na)
```

```
; Make radial range
```

```
r0=10.
```

```
r1=20.
```

```
ri=0.1
```

```
nr=ceil((r1-r0)/ri)+1
```

```
ri=(r1-r0)/(nr-1)
```

```
r=r0+ri*indgen(nr)
```

```
; Radius and azimuth for warped image
```

```
r=rebin(r,nr,na,/sample)
```

```
a=rebin(transpose(a),nr,na,/sample)
```

```
; X and Y for warped image (xc,yc is center)
```

```
xmap=xc+r*cos(a)
```

```
ymap=yc+r*sin(a)
```

```
; Warped image
```

```
oimage=Interpolate(image,xmap,ymap,/cubic)
```

```
; Median  
m=median(oimage,dim=1)
```
